

Using an Iterative DO Loop (DATA Step with a SET Statement)

The **pg2.np_summary** table contains public use statistics from the National Park Service. The Pacific West region is anticipating the number of recreational day visitors to increase yearly by 5% for national monuments and 8% for national parks. Show the forecasted number of recreational day visitors for each park for the next five years.

1. Open **p206p02.sas** from the **practices** folder. Submit the program and examine the results. Notice that the initial program is showing the forecasted value for the next year. The next year is based on adding one year to the year value of today's date.
2. Modify the program.
 - Add an iterative DO loop around the conditional IF-THEN statements.
 - The DO loop needs to iterate five times.
 - In the DO statement, a new column named **Year** needs to be created that starts at the value of **NextYear** and stops at the value of **NextYear** plus 4.
 - A row needs to be created for each year.
 - Modify the KEEP statement to keep the column **Year** instead of **NextYear**.
 - Submit the program and examine the results.

```
data ForecastDayVisits;
  set pg2.np_summary;
  where Reg='PW' and Type in ('NM','NP');
  ForecastDV=DayVisits;
  NextYear=year(today())+1;
  do Year = NextYear to NextYear+4;
    if Type='NM' then ForecastDV=ForecastDV*1.05;
    if Type='NP' then ForecastDV=ForecastDV*1.08;
    output;
  end;
  format ForecastDV comma12.;
  label ForecastDV='Forecasted Recreational Day Visitors';
  keep ParkName DayVisits ForecastDV Year;
run;

proc sort data=ForecastDayVisits;
  by ParkName;
run;

title 'Forecast of Recreational Day Visitors for Pacific West';
proc print data=ForecastDayVisits label;
run;
title;
```

3. How many rows are in the **ForecastDayVisits** table?

The **ForecastDayVisits** table has 105 rows.

4. (Optional) Modify the OUTPUT statement to be a conditional statement that outputs only on the fifth iteration. Submit the program and review the results.

```
data ForecastDayVisits;
  set pg2.np_summary;
  where Reg='PW' and Type in ('NM','NP');
  ForecastDV=DayVisits;
  NextYear=year(today())+1;
  do Year = NextYear to NextYear+4;
    if Type='NM' then ForecastDV=ForecastDV*1.05;
    if Type='NP' then ForecastDV=ForecastDV*1.08;
    if Year=NextYear+4 then output;
  end;
  format ForecastDV comma12.;
  label ForecastDV='Forecasted Recreational Day Visitors';
  keep ParkName DayVisits ForecastDV Year;
run;

proc sort data=ForecastDayVisits;
  by ParkName;
run;

title 'Forecast of Recreational Day Visitors for Pacific West';
proc print data=ForecastDayVisits label;
run;
title;
```

5. How many rows are in the output table?

There are 21 rows in the output table.