Using a Conditional DO Loop

The **pg2.np_summary** table contains public use statistics from the National Park Service. The Northeast region has seen an increase in visitors at its national monuments that previously experienced low visitation. Determine the number of years it will take for the number of visitors to exceed 100,000, assuming an annual 6% increase.

1. Open **p206p04.sas** from the **practices** folder. Submit the program and examine the results. Notice that the first two monuments are not near 100,000 visitors, but the third monument is near 100,000 after one year with a 6% increase.

2. Modify the program.

   - Add a conditional DO loop around the assignment statement where **IncrDayVisits** is being increased by 6%.
     - Add a DO UNTIL statement that executes until the value of **IncrDayVisits** exceeds 100,000.
     - Add an OUTPUT statement to show the increased values per each iteration.
     - Add an END statement.
   - Submit the program and examine the results.

```
data IncreaseDayVisits;
    set pg2.np_summary;
    where Reg='NE' and DayVisits<100000;
    IncrDayVisits=DayVisits;
    do until (IncrDayVisits>100000);
       IncrDayVisits=IncrDayVisits*1.06;
       output;
    end;
    format IncrDayVisits comma12.;
    keep ParkName DayVisits IncrDayVisits;
run;

proc sort data=IncreaseDayVisits;
    by ParkName;
run;

title1 'Years Until Northeast National Monuments Exceed 100,000 Visitors';
title2 'Based on Annual Increase of 6%';
proc print data=IncreaseDayVisits label;
    label DayVisits='Current Day Visitors'
          IncrDayVisits='Increased Day Visitors';
run;
title;
```

3. How many rows are in the **IncreaseDayVisits** table?

   The **IncreaseDayVisits** table has 41 rows.

4. Modify the program.

   - Within the DO loop, add a sum statement to add 1 to the value of **Year**.
     `Year+1;`
   - Before the DO loop, add an assignment to set the **Year** to 0.
   - Add **Year** to the KEEP statement.
   - Submit the program and examine the results.

```
data IncreaseDayVisits;
    set pg2.np_summary;
    where Reg='NE' and DayVisits<100000;
    IncrDayVisits=DayVisits;
    Year=0;
    do until (IncrDayVisits>100000);
       Year+1;
       IncrDayVisits=IncrDayVisits*1.06;
       output;
    end;
    format IncrDayVisits comma12.;
    keep ParkName DayVisits IncrDayVisits Year;
run;

proc sort data=IncreaseDayVisits;
    by ParkName;
run;

title1 'Years Until Northeast National Monuments Exceed 100,000 Visitors';
title2 'Based on Annual Increase of 6%';
proc print data=IncreaseDayVisits label;
    label DayVisits='Current Day Visitors'
          IncrDayVisits='Increased Day Visitors';
run;
title;
```

5. How many years did it take until the number of visitors exceeded 100,000 for each national monument?

   African Burial Ground National Monument: 14 years
   Booker T. Washington National Monument: 25 years
   Fort Stanwix National Monument: 2 years

6. Remove the OUTPUT statement. Submit the program and view the results. The number for **Year** should match the numbers that you specified above.

7. (Optional) Modify the DO UNTIL statement to be a DO WHILE statement that produces the same results. Submit the program and verify the results.

```
data IncreaseDayVisits;
    set pg2.np_summary;
    where Reg='NE' and DayVisits<100000;
    IncrDayVisits=DayVisits;
    Year=0;
    do while (IncrDayVisits<=100000);
       Year+1;
       IncrDayVisits=IncrDayVisits*1.06;
    end;
    format IncrDayVisits comma12.;
    keep ParkName DayVisits IncrDayVisits Year;
run;

proc sort data=IncreaseDayVisits;
    by ParkName;
run;

title1 'Years Until Northeast National Monuments Exceed 100,000 Visitors';
title2 'Based on Annual Increase of 6%';
proc print data=IncreaseDayVisits label;
    label DayVisits='Current Day Visitors'
```

```
          IncrDayVisits='Increased Day Visitors';
run;
title;
```