

Using an Iterative DO Loop with a List of Values

The **sashelp.class** table contains student information such as age, height, and weight. For the female 13-year old students, show the student's projected weight over the next four quarters, assuming a four-pound increase per quarter.

1. Open **p206p03.sas** from the **practices** folder. Modify the program. Add a DO loop to the DATA step to produce the following results. The **Weight** value is increasing by four pounds per quarter.

Projected Weight for Females (Age 13)

Obs	Name	Weight	Quarter
1	Alice	88	1
2	Alice	92	2
3	Alice	96	3
4	Alice	100	4
5	Barbara	102	1
6	Barbara	106	2
7	Barbara	110	3
8	Barbara	114	4

```
data IncWt;
  set sashelp.class;
  where Sex='F' and Age=13;
  do Quarter=1 to 4;
    Weight+4;
    output;
  end;
run;

title 'Projected Weight for Females (Age 13)';
proc print data=IncWt;
  var Name Weight Quarter;
run;
title;
```

2. Submit the program and verify the results.
3. Modify the DO statement to produce the following results. The DO statement will now be based on a list of values instead of a value that is incremented.

Projected Weight for Females (Age 13)

Obs	Name	Weight	Quarter
1	Alice	88	1st Qtr
2	Alice	92	2nd Qtr

3	Alice	96	3rd Qtr
4	Alice	100	4th Qtr
5	Barbara	102	1st Qtr
6	Barbara	106	2nd Qtr
7	Barbara	110	3rd Qtr
8	Barbara	114	4th Qtr

```

data IncWt;
  set sashelp.class;
  where Sex='F' and Age=13;
  do Quarter='1st Qtr','2nd Qtr','3rd Qtr','4th Qtr';
    Weight+4;
    output;
  end;
run;

title 'Projected Weight for Females (Age 13)';
proc print data=IncWt;
  var Name Weight Quarter;
run;
title;

```

4. Submit the program and verify the results.