Using an Iterative and Conditional DO Loop

The **pg2.eu_sports** table contains European Union trade amounts for sport products. Belgium wants to see their exports exceed their imports for golf and racket products. They expect to annually increase exports by 7% and want to achieve their goal within 10 years.

1. Open **p206p05.sas** from the **practices** folder. Submit the program and examine the results. Notice that the golf export number is farther from the golf import number as compared to the racket export and import numbers.

2. Modify the program.

    - Add a conditional DO loop around the assignment statement for **Amt_Export**.
        - Use a DO WHILE statement that executes while the export value is less than or equal to the import value.
        - Create a **Year** column that increments by a value of 1.
        - Create a row of output for each year.
    - Submit the program and examine the results.

```
data IncrExports;
    set pg2.eu_sports;
    where Year=2015 and Country='Belgium'
          and Sport_Product in ('GOLF','RACKET');
    do while (Amt_Export<=Amt_Import);
       Year+1;
       Amt_Export=Amt_Export*1.07;
       output;
    end;
    format Amt_Import Amt_Export comma12.;
run;

title 'Belgium Golf and Racket Products - 7% Increase in Exports';
proc print data=IncrExports;
    var Sport_Product Year Amt_Import Amt_Export;
run;
title;
```

3. How many years did it take until the exports exceeded the imports, and what is the final **Year** value for each sport product?

    GOLF, 14 years, 2029
    RACKET, 4 years, 2019

4. Modify the program.

    - Modify the DO statement to include an iterative portion before the conditional portion. The iterative portion needs to be based on **Year** values of 2016 to 2025 (10 years).
    - Within the DO loop, delete any statements related to the incrementing of **Year**.
    - Submit the program and review the results. The results show 14 data rows.

```
data IncrExports;
    set pg2.eu_sports;
    where Year=2015 and Country='Belgium'
          and Sport_Product in ('GOLF','RACKET');
    do Year=2016 to 2025 while (Amt_Export<=Amt_Import);
       Amt_Export=Amt_Export*1.07;
       output;
    end;
    format Amt_Import Amt_Export comma12.;
run;

title 'Belgium Golf and Racket Products - 7% Increase in Exports';
proc print data=IncrExports;
    var Sport_Product Year Amt_Import Amt_Export;
run;
title;
```

5. Complete this table based on your last modification:

Sport_Product	Number of Years	Final Year	Do Exports exceed Imports?
GOLF			
RACKET			

Sport_Product	Number of Years	Final Year	Do Exports exceed Imports?
GOLF	10	2025	No
RACKET	4	2019	Yes

6. Delete the OUTPUT statement. Submit the program and examine the results. Do these **Year** values equal the final **Year** values before deleting this statement? Why or why not?

   No, the **Year** values do not equal the final **Year** values before deleting the OUTPUT statement. Output happens after the DO loop due to the implicit OUTPUT. The **Year** column is incremented at the bottom of the DO loop before checking the DO WHILE condition at the top of the loop.

7. (Optional) Include a conditional OUTPUT statement within the DO loop that will show the two rows of output with the **Year** values equal to the final **Year** values before deleting the OUTPUT statement. Submit the program and verify the results.

```
data IncrExports;
    set pg2.eu_sports;
    where Year=2015 and Country='Belgium'
          and Sport_Product in ('GOLF','RACKET');
    do Year=2016 to 2025 while (Amt_Export<=Amt_Import);
       Amt_Export=Amt_Export*1.07;
       if Year=2025 or Amt_Export>Amt_Import then output;
    end;
    format Amt_Import Amt_Export comma12.;
run;

title 'Belgium Golf and Racket Products - 7% Increase in Exports';
proc print data=IncrExports;
```

```
    var Sport_Product Year Amt_Import Amt_Export;
run;
title;
```