Using Nested Iterative DO Loops (DATA Step with No SET Statement)

Determine the value of a retirement account after six years based on an annual investment of $10,000 and a constant annual interest rate of 7.5%.

1. Open **p206p01.sas** from the **practices** folder. Modify the program.
    - Add an iterative DO loop around the sum statement for **Invest**.
        - Add a DO statement that creates the column **Year** with values ranging from 1 to 6.
        - Add an OUTPUT statement to show the value of the retirement account for each year.
        - Add an END statement.
    - Submit the program and examine the results.

```
data retirement;
    do Year = 1 to 6;
       Invest+10000;
       output;
    end;
run;

title1 'Retirement Account Balance per Year';
proc print data=retirement noobs;
    format Invest dollar12.2;
run;
title;
```

2. What is the **Invest** value for Year 6?

    $60,000.00

3. Add an inner iterative DO loop between the sum statement and the OUTPUT statement to include the accrued quarterly compounded interest based on an annual interest rate of 7.5%.
    - Add a DO statement that creates the column **Quarter** with values ranging from 1 to 4.
    - Add a sum statement to add the accrued interest to the **Invest** value.
      `Invest+(Invest*(.075/4));`
    - Add an END statement.
    - Submit the program and examine the results.

```
data retirement;
    do Year = 1 to 6;
       Invest+10000;
       do Quarter = 1 to 4;
          Invest+(Invest*(.075/4));
       end;
       output;
    end;
run;
```

```
title1 'Retirement Account Balance per Year';
proc print data=retirement noobs;
    format Invest dollar12.2;
run;
title;
```

4. What is the **Invest** value for Year 6? Why is the **Quarter** value 5 in every row?

   $78,449.27

   The stop value for the inner DO loop is 4, so the loop will terminate when **Quarter** is equal to 5. The OUTPUT statement executes after the inner loop stops, so the value of **Quarter** is 5 in every output row.

5. Drop the **Quarter** column. Submit the program and review the results. Did the **Invest** value for Year 6 change?

```
data retirement;
    do Year = 1 to 6;
       Invest+10000;
       do Quarter = 1 to 4;
          Invest+(Invest*(.075/4));
       end;
       output;
    end;
    drop Quarter;
run;

title1 'Retirement Account Balance per Year';
proc print data=retirement noobs;
    format Invest dollar12.2;
run;
title;
```

   No, the value of **Invest** did not change when **Quarter** was dropped.